NTNU

Department of Computer Science

TDT4900 — Master's thesis

Matroids and fair allocation

Author:
Andreas Aaberge Eide

Supervisor:
Magnus Lie Hetland

March 22, 2023

Contents

Chapter 1

Introduction

Chapter 2

Background

2.1 Matroid theory

If a mathematical structure can be defined or axiomatized in multiple different, but not obviously equivalent, ways, the different definitions or axiomatizations of that structure make up a cryptomorphism. The many obtusely equivalent definitions of a matroid are a classic example of cryptomorphism, and belie the fact that the matroid is a generalization of concepts in many, seemingly disparate areas of mathematics.

First introduced by Hassler Whitney in 1935 [Whi35], in a seminal paper where he described two axioms for independence in the columns of a matrix, and defined any system obeying these axioms to be a "matroid". Whitney's key insight was that this abstraction of "independence" is applicable to both matrices and graphs. As a result of this, the terms used in matroid theory are borrowed from analogous concepts in both graph theory and linear algebra. Matroids have also been widely studied in game theory and economics, as their properties make them useful for modeling user preferences; for instance, matroid rank functions are a natural way of formally describing course allocation for students [Ben21].

Independent sets

Perhaps the most common way to define a matroid is in terms of its *independent sets*. An independence system is a pair $(E, \mathcal{I})$, where E is the ground set of

elements, $E \neq \emptyset$, and $\mathcal{I}$ is the set of independent sets, $\mathcal{I} \subseteq 2^E$. A matroid is an independence system with the following properties:

1. The empty set is independent: $\emptyset \in \mathcal{I}$.
2. The hereditary property: if $A \subseteq B$ and $B \in \mathcal{I}$, then $A \in \mathcal{I}$.
3. The augmentation property: If $A, B \in \mathcal{I}$ and $|A| > |B|$, then there exists $e \in A$ such that $B \cup \{e\} \in S$.

In practice, the ground set E represents the universe of elements in play, and the independent sets of typically represent the legal combinations of these items. In the context of fair allocation, the independent sets represent the legal (in the case of matroid constraints) or desired (in the case of matroid utilities) bundles of items.

Rank

Given a matroid $\mathfrak{M} = (E, \mathcal{I})$, the *matroid rank function* (MRF) is a function $\text{rank} : 2^E \to \mathbb{Z}^+$ that gives the *rank* of a set $A \subseteq E$, which is defined to be the size of the largest independent subset of A. Formally,

$$\text{rank}(A) = \max\{|X| : X \subseteq A \text{ and } X \in \mathcal{I}\}.$$

Matroid rank functions are *binary submodular*. Binary because they have binary marginals, that is, $\text{rank}(A \cup \{e\}) - \text{rank}(A) \in \{0, 1\}$, for all $A \subseteq 2^E$ and $e \in E$. Submodularity refers to rank functions' natural diminishing returns property, namely that for any two sets $X, Y \subseteq E$, we have

$$\text{rank}(X \cup Y) + \text{rank}(X \cap Y) \leq \text{rank}(X) + \text{rank}(Y).$$

This diminishing returns property makes the rank function useful for modeling user preferences, and is a reason why matroids show up so often in economics and game theory (???).

Closed sets

We also need to establish the concept of the *closed sets* of a matroid. A closed set is a set whose cardinality is maximal for its rank. Equivalently to the definition given above, we can define a matroid as $\mathfrak{M} = (E, \mathcal{F})$, where $\mathcal{F}$ is the set of closed sets of $\mathfrak{M}$, satisfying the following properties:

1. The set of all elements is closed: $E \in \mathcal{F}$
2. The intersection of two closed sets is a closed set: If $A, B \in \mathcal{F}$, then $A \cap B \in \mathcal{F}$
3. If $A \in \mathcal{F}$ and $a, b \in E \setminus A$, then b is a member of all sets in $\mathcal{F}$ containing $A \cup \{a\}$ if and only if a is a member of all sets in $\mathcal{F}$ containing $A \cup \{b\}$

2.2 Examples of matroids

The free matroid

The uniform matroid

The vector matroid

The graphic matroid

Chapter 3

Random matroid generation

One goal for this project is to create the Julia library `Matroids.jl`, which will supply functionality for generating and interacting with random matroids. In the preparatory project delivered fall of 2022, I implemented Knuth's 1974 algorithm for the random generation of arbitrary matroids via the erection of closed sets [Knu75]. With this, I was able to randomly generate matroids with a universe size n of about 12, but for larger values of n my implementation was unbearably slow. In this chapter, Knuth's method for random matroid construction will be described, along with the steps I have taken to speed up my initial, naïve implementation. The random generation of other specific types of matroids is discussed as well.

3.1 Knuth's matroid construction (KMC)

KNUTH-MATROID (given in Algorithm 1) accepts the ground set E and a list of enlargements X, and produces the matroid over E where each set in $\mathrm{X}[r]$ is a closed set of rank r. The output is the list $\mathrm{F} = [F_0, \ldots, F_r]$, where r is the final rank of $\mathfrak{M}$ and F_i is the set of closed sets of rank i. In the paper, Knuth shows that $\bigcup_{i=0}^{r} \mathrm{F}[r] = \mathcal{F}$, and so the resulting structure $\mathfrak{M} = (E, \mathcal{F})$ is a matroid.

The algorithm proceeds in a bottom-up manner, starting with the single closed set of rank 0 (the empty set) and for each rank $r + 1$ adds the covers of the closed sets of rank r. The covers of a closed set A of rank r is simply all sets obtained by adding one more element from E to A. The covers are generated with the helper method GENERATE-COVERS(F, r, E).

```
GENERATE-COVERS(F, r, E)
1  return {A ∪ {a} : A ∈ F[r], a ∈ E \ A}
```

Given no enlargements (X = []), the resulting matroid is the uniform matroid of rank $|E|$. Arbitrary matroids can be generated by supplying different lists X. When enlarging, the sets in X$[r+1]$ are simply added to F$[r+1]$.

SUPERPOSE!(F$[r+1]$, $F[r]$) ensures that the newly enlarged set of closed sets of rank $r+1$ is valid. If F_{r+1} contains two sets A, B whose intersection $A \cap B \not\subseteq C$, for some $C \in F_r$, replace A, B with $A \cup B$. Repeat until no two sets exist in F_{r+1} whose intersection is not contained within some set $C \in F_r$.

```
SUPERPOSE!(F_{r+1}, F_r)
 1  for A ∈ F_{r+1}
 2      for B ∈ F_{r+1}
 3          flag ← true
 4          for C ∈ F_r
 5              if A ∩ B ⊆ C
 6                  flag ← false
 7
 8          if flag = true
 9              F_{r+1} ← F_{r+1} \ {A, B}
10              F_{r+1} ← F_{r+1} ∪ {A ∪ B}
```

3.1.1 Randomized KMC

In the randomized version of KNUTH-MATROID, we generate matroids by applying a supplied number of random coarsening steps, instead of enlarging with supplied sets. This is done by applying SUPERPOSE! immediately after adding the covers, then choosing a random member A of F$[r+1]$ and a random element $a \in E \setminus A$, replacing A with $A \cup \{a\}$ and finally reapplying SUPERPOSE!. The parameter $p = (p_1, p_2, \ldots)$ gives the number of such coarsening steps to be applied at each iteration of the algorithm.

The pseudocode given up to this point corresponds closely to the initial Julia implementation, which can be found in Appendix A.1. It should already be clear that this brute force implementation leads to poor performance – for instance, the SUPERPOSE! method uses a triply nested for loop, which is a candidate

Algorithm 1 KNUTH-MATROID(E, X)

Input: The ground set of elements E, and a list of enlargements X.

Output: The list of closed sets of the resulting matroid grouped by rank, $\mathrm{F} = [F_0, \ldots, F_r]$, where F_i is the set of closed sets of rank i.

```
1  r = 0, F = [{∅}]
2  while true
3      PUSH!(F, GENERATE-COVERS(F, r, E))
4      F[r + 1] = F[r + 1] ∪ X[r + 1]
5      SUPERPOSE!(F[r + 1], F[r])
6      if E ∉ F[r + 1]
7          r ← r + 1
8      else
9          return (E, F)
```

for significant improvement if possible. Section 3.1.2 describes the engineering work done to create a more performant implementation.

3.1.2 Improving performance

When recreating Knuth's table of observed mean values for the randomly generated matroids, some of the latter configurations of n and $(p_1, p_2, \ldots)$ was unworkably slow, presumably due to the naïve implementation of the algorithm. Table 3.1 shows the performance of this first implementation.

The performance was measured using Julia's `@timed`[1] macro, which returns the time it takes to execute a function call, how much of that time was spent in garbage collection and the size of the memory allocated. As is evident from the data, larger matroids are computationally quite demanding to compute with the current approach, and the time and space requirements scales exponentially with n. Can we do better? As it turns out, we can; after the improvements outlined in this section, we will be able to generate matroids over universes as large as $n = 128$ in a manner of seconds and megabytes.

[1] https://docs.julialang.org/en/v1/base/base/#Base.@timed

Algorithm 2 RANDOMIZED-KNUTH-MATROID(E, p)

Input: The ground set of elements E, and a list $p = [p_1, p_2, ...]$, where p_r is the number of coarsening steps to apply at rank r in the construction.

Output: The list of closed sets of the resulting matroid grouped by rank, $\mathrm{F} = [F_0, \ldots, F_r]$, where F_i is the set of closed sets of rank i.

$r = 0, \mathrm{F} = [\{\emptyset\}]$
while true
    PUSH!$(\mathrm{F}, $GENERATE-COVERS$(\mathrm{F}, r, E))$
    SUPERPOSE!$(\mathrm{F}[r+1], \mathrm{F}[r])$
    if $E \in \mathrm{F}[r+1]$ **return** (E, F)
    while $p[r] > 0$
        let A **be a random set in** $\mathrm{F}[r+1]$
        let a **be a random element in** $E \setminus A$
        replace A **with** $A \cup \{a\}$ **in** $\mathrm{F}[r+1]$
        SUPERPOSE!$(\mathrm{F}[r+1], \mathrm{F}[r])$
        if $E \in \mathrm{F}[r+1]$ **return** (E, F)
        $p[r] -= 1$
    $r += 1$

Representing sets as binary numbers

The first improvement we will attempt is to represent our families as sets of hexadecimal numbers, instead of sets of sets of numbers. Sets are represented using Julia's native `Set` type [2]. The Julia implementation at this point can be found in Appendix A.2.

[2] https://docs.julialang.org/en/v1/base/collections/#Base.Set

Table 3.1: Performance of `random_kmc_v1`.

n	$(p_1, p_2, \ldots)$	Trials	Time	GC Time	Bytes allocated
10	(0, 6, 0)	100	0.0689663	0.0106786	147.237 MiB
10	(0, 5, 1)	100	0.1197194	0.0170734	251.144 MiB
10	(0, 5, 2)	100	0.0931822	0.0144022	203.831 MiB
10	(0, 6, 1)	100	0.0597314	0.0094902	132.460 MiB
10	(0, 4, 2)	100	0.1924601	0.0284532	406.131 MiB
10	(0, 3, 3)	100	0.3196838	0.0463972	678.206 MiB
10	(0, 0, 6)	100	1.1420602	0.1671325	2.356 GiB
10	(0, 1, 1, 1)	100	2.9283978	0.3569357	5.250 GiB
13	(0, 6, 0)	10	104.0171128	9.9214449	161.523 GiB
13	(0, 6, 2)	10	11.4881308	1.3777947	20.888 GiB
16	(6, 0, 0)	1	-	-	-

The idea is to define a family of closed sets of the same rank as `Set{UInt16}`. Using `UInt16` we can support ground sets of size up to 16. Each 16-bit number represents a set in the family. For instance, the set $\{2, 5, 7\}$ is represented by

$$164 = \text{0x00a4} = \text{0b0000000010100100} = 2^7 + 2^5 + 2^2.$$

At either end we have $\emptyset \equiv$ 0x0000 and $E \equiv$ 0xffff (if $n = 16$). Set operations have equivalent binary operations; intersection corresponds to bitwise AND, union to bitwise OR and the set difference between sets A and B to the bitwise OR of A and the complement of B. Subset equality is also simple to implement: $A \subseteq B \iff A \cap B = A$.

We can now describe the bitwise versions of the required methods. The bitwise implementation of GENERATE-COVERS finds all elements in $E \backslash A$ by finding each value i for which `A & 1 << i === 0`, meaning that the set represented by `1 << i` is not a subset of A. The bitwise implementation of SUPERPOSE! is unchanged apart from using the bitwise set operations described above.

The performance of `random_kmc_v2` is shown in Table 3.2. It is clear that representing closed sets using binary numbers represents a substantial im-

Table 3.2: Performance of `random_kmc_v2`.

n	$(p_1, p_2, \ldots)$	Trials	Time	GC Time	Bytes allocated
10	[0, 6, 0]	100	0.0010723	0.0001252	1.998 MiB
10	[0, 5, 1]	100	0.0017543	0.0001431	3.074 MiB
10	[0, 5, 2]	100	0.0008836	0.0001075	2.072 MiB
10	[0, 6, 1]	100	0.0007294	6.73e-5	1.700 MiB
10	[0, 4, 2]	100	0.0020909	0.0001558	3.889 MiB
10	[0, 3, 3]	100	0.0024636	0.0002139	4.530 MiB
10	[0, 0, 6]	100	0.007082	0.0004801	9.314 MiB
10	[0, 1, 1, 1]	100	0.0132477	0.0008307	17.806 MiB
13	[0, 6, 0]	10	0.042543	0.0014988	31.964 MiB
13	[0, 6, 2]	10	0.0183313	0.0012176	21.062 MiB
16	[0, 6, 0]	10	1.2102877	0.0146129	450.052 MiB

provement – we are looking at performance increases of 100x-1000x across the board.

Sorted superpose

Can we improve the running time of the algorithm further? It is clear that SUPERPOSE! takes up a large portion of the compute time. In the worst case, when no enlargements have been made, F_{r+1} is the set of all $r+1$-sized subsets of E, $|F_{r+1}| = \binom{n}{r+1}$. Comparing each $A, B \in F_{r+1}$ with each $C \in F_r$ in a triply nested for loop requires a whopping $\mathcal{O}(\binom{n}{r+1}^2 \binom{n}{r})$ operations. In the worst case, no enlargements are made at all, and we build the free matroid in $\mathcal{O}(2^{3n})$ time.

Observing the step-by-step calls to SUPERPOSE!, we see that there are a lot of duplicate and unnecessary sets being processed. The duplicate sets stem from GENERATE-COVERS, whose current implementation does not take into account that any two sets of rank r will have at least one cover in common. To see this, consider a matroid-under-construction where $A = \{1, 2\}$ and $B = \{1, 3\}$ are closed sets of rank 2. Currently, GENERATE-COVERS will happily generate the cover $C = \{1, 2, 3\}$ twice, once as the cover of A and subsequently as the cover of B. Non-redundant cover generation was a worthwhile improvement and is described below.

After larger closed sets have been added to $\mathrm{F}[r+1]$, SUPERPOSE! will cause sets to merge, so that only maximal dependent sets remain. Some sets will even simply disappear. In the case where $X = \{1, 2\}$ was added by GENERATE-COVERS, and the $Y = \{1, 2, 3\}$ was added manually as an enlargement, the smaller set will be completely subsumed in the bigger set, as $\{1, 2\} \cap \{1, 2, 3\} = \{1, 2\}$ and $\{1, 2\} \cup \{1, 2, 3\} = \{1, 2, 3\}$. In this situation, X will "eat" the covers $\{1, 3\}$ and $\{2, 3\}$ as well. Since the larger sets will absorb so many of the smaller sets (around $\binom{p}{r+1}$, where p is the size of the larger set and $r + 1$ is the size of the smallest sets allowed to be added in a given iteration), might it be an idea to perform the superpose operation in descending order based on the size of the sets? This should result in fewer calls to SUPERPOSE!, as the bigger sets will remove the smaller sets that fully overlap with them in the early iterations, however, the repeated sorting of the sets might negate this performance gain. This was the idea behind `random_kmc_v3`, which can be found in Appendix A.2.

Unfortunately, as Table 3.3 shows, this implementation is a few times slower and more space demanding than the previous implementation. This is likely due to the fact that an ordered list is more space inefficient than the hashmap-based `Set`.

Table 3.3: Performance of `random_kmc_v3`.

n	$(p_1, p_2, \ldots)$	Trials	Time	GC Time	Bytes allocated
10	[0, 6, 0]	100	0.0023382	0.0001494	4.042 MiB
10	[0, 5, 1]	100	0.001853	0.0001433	4.383 MiB
10	[0, 5, 2]	100	0.0017845	0.0001341	4.043 MiB
10	[0, 6, 1]	100	0.0015145	0.0001117	3.397 MiB
10	[0, 4, 2]	100	0.0030704	0.0002125	6.385 MiB
10	[0, 3, 3]	100	0.0037838	0.0002514	7.018 MiB
10	[0, 0, 6]	100	0.008903	0.000557	14.159 MiB
10	[0, 1, 1, 1]	100	0.0142828	0.0008823	21.838 MiB
13	[0, 6, 0]	10	0.0627633	0.002094	51.492 MiB
13	[0, 6, 2]	10	0.0106478	0.0007704	20.774 MiB
16	[0, 6, 0]	10	0.6070136	0.0095656	310.183 MiB

```
                                                                                2
```

Iterative superpose

So far, we are inserting all covers of F_r into F_{r+1} along with all enlargements, and then running the superpose operation on all of them at once. Until this point, the superpose operation was performed with a triply nested `for` loop, c Thus we were looking at a whopping to perform the superpose part of step r.

Table 3.4: Performance of `randomized_kmc_v4`.

Rank table

Non-redundant cover generation and iterative superpose

3.1.3 Finding independent sets and circuits

In order for the

3.2 Other kinds of matroids

3.2.1 Uniform matroids

3.2.2 Graphic matroids

3.2.3 Vector matroids

Chapter 4

A library for fair allocation with matroids

A goal for this project is to introduce Matroids.jl as a useful library for experimenting with fair allocation that require matroids. In the previous chapter, we explored how this library generates random matroids, however this is not very useful until we also have in place an API layer to allow fair allocation algorithms to interface with our matroids in a practical and efficient manner.

4.1 The matroid union algorithm

4.2 Supporting universe sizes of $n > 128$

The larger the ground set, the closer we are to an instance of The cake-cutting problem. Typical fair allocation problems with indivisible items deal with less

than 100 items.

In other words, the Integer cap of 128 bits is a reasonable upper limit on universe size for fair allocation problems. However, one could look into using packages that add larger fixed-width integer types[1]. `Matroids.jl` supports arbitrary integer types.

[1] See for instance `BitIntegers.jl`

Chapter 5

Fair allocation with matroid rank utilities

We wish to study fair allocation in which the utility function of each agent is a matroid rank function. That is, the utility function for each agent $i \in [n]$ is the rank function for a matroid $\mathfrak{M}_i = ([m], \mathcal{I}_i)$.

$S_i \in \mathcal{I}_i \iff v_i(S_i) = |S_i|$.

5.1 Yankee Swap

5.2 Barman and Verma's MMS algorithm

Chapter 6

Fair allocation under matroid constraints

6.1 Gourvès and Monnot's MMS approximation algorithm

6.2 Biswas and Barman's algorithm for EF1 under cardinality constraints

6.3 Biswas and Barman's algorithm for EF1 under matroid constraints

Chapter 7

Results

Chapter 8

Conclusions and future work

Notes

1. Skrive mer om hvordan Set{Set{Integer}} lagres i minnet og fordelene med å gå over til Set{Integer}.
2. Skrive om variansen mellom tilfeldige matroider! @benchmark osv. Histogram
3. Referer til Spliddit og vanlige størrelser på fordelingsproblemer
4. Beskriv åssen man kan oppgi valgfri Integer-type

Bibliography

[Ben21] et al. Benabbou, Nawal. Finding fair and efficient allocations for matroid rank valuations. *ACM Transactions on Economics and Computation*, 9:1–41, December 2021.

[Knu75] Donald E. Knuth. Random matroids. *Discrete Mathematics*, 12:341–358, 1975.

[Whi35] Hassler Whitney. On the abstract properties of linear dependence. *American Journal of Mathematics*, 57:509–533, July 1935.

Appendices

Appendix A

Code snippets

A.1 `random_kmc_v1`

```
"""
Generate the set F_{r+1} of all "covers" of the sets in F_r, given the ground
     set of elements E. Set-based.
"""
function generate_covers_v1(Fr, E)
  Set([A ∪ a for A ∈ Fr for a ∈ setdiff(E, A)])
end


"""
  If F_{r+1} contains any two sets A, B whose intersection A ∩ B is not
       contained in C for any C ∈ F_r, replace A, B ∈ F_{r+1} by the single set
       A ∪ B. Repeat this operation until A ∩ B ⊆ C for some C ∈ F_r whenever A
       and B are distinct members of F_{r+1}.

  F and F_old should be Family: A Set of Sets of some type.

  This is the first, Set-based implementation of this method.
"""
function superpose_v1!(F, F_old)
  for A ∈ F
    for B ∈ F
      should_merge = true
      for C ∈ F_old
        if A ∩ B ⊆ C
          should_merge = false
        end
      end
```

```
      if should_merge
        setdiff!(F, [A, B])
        push!(F, A ∪ B)
      end
    end
  end

  return F
end

"""
First implementation of Knuth's random matroid construction through random "
     coarsening".

n is the size of the universe.
p is a list (p_1, p_2, ...), where p_r is the number of coarsening steps to
     apply at rank r in the construction. The first entry of p should usually be
      0, since adding closed sets of size > 1 at rank 1 is equivalent to
     shrinking E.

This uses the Set-based KMC methods.
"""
function random_kmc_v1(n, p, T)::KnuthMatroid{Set{Integer}}
  E = Set([i for i in range(0,n-1)])

  # Step 1: Initialize.
  r = 1
  F = [family([])]
  pr = 0

  while true
    # Step 2: Generate covers.
    push!(F, generate_covers_v1(F[r], E))

    # Step 4: Superpose.
    superpose_v1!(F[r+1], F[r])

    # Step 5: Test for completion.
    if E ∈ F[r+1]
      return KnuthMatroid{Set{Integer}}(n, F, [], Set(), Dict())
    end

    if r <= length(p)
      pr = p[r]
    end

    while pr > 0
      # Random closed set in F_{r+1} and element in E \ A.
      A = rand(F[r+1])
      a = rand(setdiff(E, A))

      # Replace A with A ∪ {a}.
      F[r+1] = setdiff(F[r+1], A) ∪ Set([A ∪ a])

      # Superpose again to account for coarsening step.
      superpose_v1!(F[r+1], F[r])
```

```
      # Step 5: Test for completion.
      if E ∈ F[r+1]
        return KnuthMatroid{Set{Integer}}(n, F, [], Set(), Dict())
      end

      pr -= 1
    end


    r += 1
  end
end
```

A.2 `random_kmc_v2` and `random_kmc_v3`

```
"""
Generate the set F_{r+1} of all "covers" of the sets in F_r, given the size of
     the universe. Bit-based.
"""
function generate_covers_v2(F_r, n)
  Set([A | 1 << i for A ∈ F_r for i in 0:n-1 if A & 1 << i === 0])
end

"""
Returns whether the intersection of A and B is contained within
some C in F_prev.
"""
function should_merge(A, B, F_prev)
  for C in F_prev
    if subseteq(A & B, C)
      return false
    end
  end
  return true
end

"""
If F contains any two sets A, B whose intersection A ∩ B is not contained in C
     for any C ∈ F_prev, replace A, B ∈ F with the single set A ∪ B. Repeat this
      operation until A ∩ B ⊆ C for some C ∈ F_prev whenever A and B are
     distinct members of F.

This implementation represents the sets using bits.
"""
function bitwise_superpose!(F, F_prev)
  As = copy(F)
  while length(As) !== 0
    A = pop!(As)

    for B in setdiff(F, A)
```

```
      if should_merge(A, B, F_prev)
        push!(As, A | B)
        setdiff!(F, [A, B])
        push!(F, A | B)
        break
      end
    end
  end

  return F
end

function sorted_bitwise_superpose!(F, F_prev)
  As = sort!(collect(F), by = s -> length(bits_to_set(s)))
  while length(As) !== 0
    A = popfirst!(As)

    for B in setdiff(F, A)
      if should_merge(A, B, F_prev)
        insert!(As, 1, A | B)
        setdiff!(F, [A, B])
        push!(F, A | B)
        break
      end
    end
  end

  return F
end

"""
Bitwise implementation of Knuth's approach to random matroid generation through
     a number of random "coarsening" steps. Supply the generate_covers and
     superpose methods to study the effects of different implementations of
     these.

n is the size of the universe.
p is a list (p_1, p_2, ...), where p_r is the number of coarsening steps to
     apply at rank r in the construction. The first entry of p should usually be
      0, since adding closed sets of size > 1 at rank 1 is equivalent to
     shrinking E.
"""
function random_bitwise_kmc(generate_covers, superpose, n, p)::KnuthMatroid{Any}
  # Initialize.
  r = 1
  pr = 0
  F = [Set(0)]
  E = 2^n - 1 # The set of all elements in E.

  while true
    # Generate covers.
    push!(F, generate_covers(F[r], n))

    # Superpose.
    superpose(F[r+1], F[r])

    # Test for completion.
    if E ∈ F[r+1]
```

```
      return KnuthMatroid{Any}(n, F, [], Set(), Dict())
    end

    # Apply coarsening.
    if r <= length(p)
      pr = p[r]
    end

    while pr > 0
      # Get random closed set A in F_{r+1} and element a in E - A.
      A = rand(F[r+1])
      a = random_element(diff(E, A))

      # Replace A with A ∪ {a}.
      F[r+1] = setdiff(F[r+1], A) ∪ Set([A | a])

      # Superpose again to account for coarsening step.
      superpose(F[r+1], F[r])

      # Step 5: Test for completion.
      if E ∈ F[r+1]
        return KnuthMatroid{Any}(n, F, [], Set(), Dict())
      end

      pr -= 1
    end

    r += 1
  end
end

"""
Second implementation of random-KMC. This uses the bit-based KMC methods.
"""
function random_kmc_v2(n, p, T=UInt16)
  return random_bitwise_kmc(generate_covers_v2, bitwise_superpose!, n, p)
end

"""
Third implementation of random-KMC. This sorts the sets by size before
    superposing.
"""
function random_kmc_v3(n, p, T=UInt16)
  return random_bitwise_kmc(generate_covers_v2, sorted_bitwise_superpose!, n, p)
end
```

A.3 `random_kmc_v4`

```
"""
This is an attempt at a smarter implementation than directly following the setup
     from Knuth's 1974 article. The superpose step is replaced by an insert
```

```
    operation that inserts new closed sets into the family of current rank one
    at a time, superposing on the fly.
"""
function randomized_knuth_matroid_construction_v4(n, p, T=UInt16)::KnuthMatroid{
    T}
  r = 1
  pr = 0
  F = [Set(0)]
  E = 2^n - 1 # The set of all elements in E.

  while true
    to_insert = generate_covers_v2(F[r], n)

    # Apply coarsening to covers.
    if r <= length(p) && E ∉ to_insert # No need to coarsen if E is added.
      pr = p[r]
      while pr > 0
        A = rand(to_insert)
        a = random_element(E - A)
        to_insert = setdiff(to_insert, A) ∪ [A | a]
        pr -= 1
      end
    end

    # Superpose.
    push!(F, Set()) # Add F[r+1].
    while length(to_insert) > 0
      A = pop!(to_insert)
      push!(F[r+1], A)

      for B in setdiff(F[r+1], A)
        if should_merge(A, B, F[r])
          push!(to_insert, A | B)
          setdiff!(F[r+1], [A, B])
          push!(F[r+1], A | B)
        end
      end
    end

    if E ∈ F[r+1]
      return KnuthMatroid{T}(n, F, [], Set(), Dict())
    end

    r += 1
  end
end
```

A.4 random_kmc_v5

```
"""
```

```
Fifth implementation of random-KMC. This one uses a dictionary to keep track of
    previously seen sets.
"""
function random_kmc_v5(n, p, T=UInt16)::KnuthMatroid{T}
  r = 1
  pr = 0
  F::Vector{Set{T}} = [Set(T(0))]
  E = 2^n-1
  rank = Dict{T, UInt8}(0=>0) # The rank table maps from the representation of a
      set to its assigned rank.

  while true
    to_insert = generate_covers_v2(F[r], n)

    # Apply coarsening to covers.
    if r <= length(p)
      pr = p[r]
      while length(to_insert) > 0 && pr > 0 && E ∉ to_insert # No need to
          coarsen if E is added.
        A = rand(to_insert)
        a = random_element(E - A)
        to_insert = setdiff(to_insert, A) ∪ [A | a]
        pr -= 1
      end
    end

    # Superpose.
    push!(F, Set()) # Add F[r+1].
    while length(to_insert) > 0
      A = pop!(to_insert)
      push!(F[r+1], A)
      rank[A] = r

      for B in setdiff(F[r+1], A)
        if !haskey(rank, A&B) || rank[A&B] >= r
          # Update insert queue.
          push!(to_insert, A | B)

          # Update F[r+1].
          setdiff!(F[r+1], [A, B])
          push!(F[r+1], A | B)

          # Update rank table.
          rank[A|B] = r
          break
        end
      end
    end

    if E ∈ F[r+1]
      return KnuthMatroid{T}(n, F, [], Set(), rank)
    end

    r += 1
  end
end
```

A.5 random_kmc_v6

```
"""
Sixth implementation of random-KMC, in which a rank table is used to keep track
     of set ranks, and the covers and enlargements are added one at a time,
     ensuring the matroid properties at all times.
"""
function random_kmc_v6(n, p, T=UInt16)::KnuthMatroid{T}
  r = 1
  pr = 0
  F::Vector{Set{T}} = [Set(T(0))]
  E::T = BigInt(2)^n-1
  rank = Dict{T, UInt8}(0=>0)

  while E ∉ F[r]
    # Create empty set.
    push!(F, Set())

    # Generate minimal closed sets for rank r+1.
    for y in F[r] # y is a closed set of rank r.
      t = E - y # The set of elements not in y.
      # Find all sets in F[r+1] that already contain y and remove excess
          elements from t.
      for x in F[r+1]
        if (x & y == y) t &= ~x end
      end
      # Insert y ∪ a for all a ∈ t.
      while t > 0
        x = y|(t&-t)
        add_set!(x, F, r, rank)
        t &= ~x
      end
    end

    if E ∈ F[r+1]
      break
    end

    if r <= length(p)
      # Apply coarsening.
      pr = p[r]
      while pr > 0 && E ∉ F[r+1]
        A = rand(F[r+1])
        t = E-A
        one_element_added::Vector{T} = []
        while t > 0
          x = A|(t&-t)
          push!(one_element_added, x)
          t &= ~x
        end
        Acupa = rand(one_element_added)
        setdiff!(F[r+1], A)
        add_set!(Acupa, F, r, rank)
        pr -= 1
      end
```

```
    end

    r += 1
  end

  return KnuthMatroid{T}(n, F, [], Set(), rank)
end

function add_set!(x, F, r, rank)
  if x in F[r+1] return end
  for y in F[r+1]
    if haskey(rank, x&y) && rank[x&y]<r
      continue
    end

    # x ∩ y has rank > r, replace with x ∪ y.
    setdiff!(F[r+1], y)
    return add_set!(x|y, F, r, rank)
  end

  push!(F[r+1], x)
  rank[x] = r
end
```